
//*******************************************************************
//Name:MASTERCON.WCM
//Directory:Office\Prwin\Macros
//Description:Converts Prwin 2.0 masters to Prwin 3.0 masters.
//*******************************************************************
Application (A1; "WordPerfect"; Default; "US") 
Application (A2; "WPPrWin"; Default; "US")
//*******************************************************************
//MAIN
//*******************************************************************

CANCEL(On!)
ONCANCEL CALL(DestroyInitDialog)
CALL(InitializeVariables)
CALL(CreateInitDialog)
CALL(DisplayDialog)
CALL(StartInitMessageLoop)
CALL(CreateListDialog)
CALL(DisplayListDialog)
CALL(StartListMessageLoop)
CALL(QuitMacro)

LABEL(InitializeVariables)
WM_SYSCOMMAND := 274
TempMaster := ""
vStringLen := 0
vTotalLen := 0
vDirLen := 0
vDirLen1 := 0
vLoop := "Yes"
vListLoop := "Yes"
vListItem :=""
vListItem1 := ""
vDir := ""
vDir1 := ""
y := 1
z := 1
HdReturn := NTOC(0F90Ah)
DECLARE (Add[256])
DECLARE (Target[256])
NoSelectAll := FALSE
RETURN

LABEL(CreateInitDialog)
DIALOGDEFINE(1000;50;50;210;115;1+2;"Master Conversion")
DIALOGADDFILENAMEBOX(1000;101;40;20;150;15;DirOnly!;vDir;"c:\";"*.mst")
DIALOGADDFILENAMEBOX(1000;102;40;40;150;15;DirOnly!;vDir1;"c:\";"*.mst")
DIALOGADDTEXT(1000;103;5;23;30;15;1;"Source:")
DIALOGADDTEXT(1000;104;5;43;30;15;1;"Target:")
RETURN

LABEL(CreateListDialog)
DIALOGDEFINE(1001;50;50;210;175;2;"Master Files")
DIALOGADDPUSHBUTTON(1001;201;125;140;35;13;0;"Convert")
DIALOGADDPUSHBUTTON(1001;202;165;11;35;13;0;"ADD")



DIALOGADDPUSHBUTTON(1001;207;165;28;35;13;0;"Select All")
DIALOGADDLISTBOX(1001;203;5;12;150;60;1;vListIem)
DIALOGADDLISTBOX(1001;204;5;85;150;55;1;vListIem1)
DIALOGADDTEXT(1001;205;5;1;150;10;1;"Masters Found:   " + vDir)
DIALOGADDTEXT(1001;206;5;73;150;10;1;"Masters to Convert:  " + vDir1 )
RETURN

LABEL(DisplayDialog)
DIALOGDISPLAY(1000;101;MsgLoop)
RETURN

LABEL(DisplayListDialog)
DIALOGDISPLAY(1001;203;MsgLoopList)
vCtrl := 1
FORNEXT(vCtrl;1;z - 1;1)

DIALOGADDLISTITEM(1001;203;Add[vCtrl])
ENDFOR
RETURN

LABEL(DisplayTargetListControl)
vListCtrl := 1
FORNEXT(vListCtrl;1;z - 1;1)

DIALOGADDLISTITEM(1001;204;Add[vListCtrl])
Target[vListCtrl] := Add[vListCtrl]
y := y + 1

ENDFOR
RETURN

LABEL(StartInitMessageLoop)
WHILE(vLoop = "Yes")
ENDWHILE
RETURN

LABEL(StartListMessageLoop)
WHILE(vListLoop = "Yes")
ENDWHILE
RETURN

LABEL(RemovePath)
z := x
x := 1
WHILE(Add[x] != "")

vStringLen := STRLEN(Add[x])
vTotalLen := vStringLen - vDirLen 
Add[x] := SUBSTR(Add[x];vDirLen + 1;vTotalLen) 
x := x + 1

ENDWHILE
RETURN

LABEL(BuildList)
x := 1
Add[x] = FileFind( vDir +"*.mst";;1)

while (Add[x] != "")
x := x + 1
Add[x] := FileFind("";;1)

endwhile
CALL(RemovePath)



RETURN

LABEL (GetPath)
vDirLen := STRLEN(vdir) 
vBackSlashSource := SUBCHAR(vDir;vDirLen ;1)

IF(vBackSlashSource  != "\")
vDir := vDir + "\"
vDirLen := vDirLen + 1

ENDIF
ONCANCEL(DestroyInitDialog)
IF(DOESDIRECTORYEXIST(vDir)!=TRUE)

PROMPT("Directory Not Found"; "Directory entered could not be found, do 
you wish to create it   "  + vDir)

PAUSE
CREATEDIRECTORY(vdir)

ENDIF

vDirLen1 := STRLEN(vDir1)
vBackSlashTarget := SUBCHAR(vDir1;vDirLen1 ;1)

IF(vBackSlashTarget  != "\")
vDir1 := vDir1 + "\"
vDirLen1 := vDirLen1 + 1

ENDIF
IF(DOESDIRECTORYEXIST(vDir1)!=TRUE)

PROMPT("Directory Not Found"; "Directory entered could not be found, do 
you wish to create it   "  + vDir1)

PAUSE
CREATEDIRECTORY(vdir1)

ENDIF

RETURN

LABEL(AddDuplicateMasters)
m := 1
vCompare := 1
FORNEXT(vCompare;1;y - 1;1)

IF(TempMaster = Target[vCompare])
MESSAGEBOX(vStatus;"Duplicate Master";"The Master selected has already been 

added to the list   " + Target[vCompare] )
vCompare := y

ELSE
m := m + 1

ENDIF
ENDFOR

IF(m = y)
Target[y] := TempMaster
DIALOGADDLISTITEM(1001;204; Target[y])
y := y + 1

ENDIF
RETURN

LABEL(DestroyInitDialog)
DIALOGDESTROY(1000)
QUIT

LABEL(QuitMacro)



DIALOGDESTROY(1001)
QUIT

LABEL(MsgLoop)
SWITCH(MsgLoop[3])

CASEOF "CancelBttn":
vLoop := "No"
DIALOGDESTROY(1000)
QUIT

CASEOF "OKBttn":
RegionGetWindowText (vDir; 1000.101)
RegionGetWindowText (vDir1; 1000.102)
IF((STRLEN(vDir) = 0)OR(STRLEN(vDir1)=0))

MESSAGEBOX(vStatus;"File Error";"You must enter text in both edit controls 
" )

ELSE
CALL(GetPath)
CALL(BuildList)
vLoop := "No"
DIALOGDESTROY(1000)

ENDIF
ENDSWITCH
IF(MsgLoop[5] = WM_SYSCOMMAND)

DIALOGDESTROY(1000)
QUIT

ENDIF
RETURN

LABEL(MsgLoopList)
IF(MsgLoopList[5] = WM_SYSCOMMAND) //Checks for CLOSE command

CALL(QuitMacro) //Terminates macro on CLOSE
ENDIF
SWITCH(MsgLoopList[3])

CASEOF 201: //Convert button
FORNEXT(vFile;1;y-1;1)

FileOpen(vDir+ Target[vFile])
InhibitInput(Off!)
SENDKEYS(vDir1+Target[vFile] + "{Enter}")
FileSaveAsDlg()
FileClose ()

ENDFOR
vListLoop := "No"

CASEOF 202: //Add button
REGIONGETSELECTEDTEXT(TempMaster;1001.203)
IF(y = 1)

Target[1] := TempMaster
DIALOGADDLISTITEM(1001;204; Target[y])
y := y + 1

ELSE
CALL(AddDuplicateMasters)

ENDIF

CASEOF 207: //Select All button
IF(NoSelectAll = FALSE)

REGIONRESETLIST("1001.204")
CALL(DisplayTargetListControl)



ELSE
MESSAGEBOX(vStatus;"Select All";"Select All can only be used one time")

ENDIF
NoSelectAll := TRUE

CASEOF "CancelBttn": //Cancel button
vListLoop := "No"

ENDSWITCH
RETURN 


